
1 Introduction

This note discusses some practical aspects of specification that are not ex-
plained clearly or not explained at all in the TLA+ book [2]. I take as an
example a very simple lock API that is specified in Section 2. Section 3
discusses the concept of an interface and explains how the specification of
the lock API’s interface can be put into a separate module. Section 4 gives
a simple implementation of the lock API and explains how we show the
implementation’s correctness. Section 5 explains how to use the lock API’s
specification in the specification of a system that uses that API.

2 The Example Specification

We now give our example lock API specification, which is formula LockSpec
of the following module. The specification is simple and is explained by the
comments. The rest of this note assumes that you understand the purpose
of the declarations and definitions; but you don’t have to understand the
specification in complete detail.

module Lock1
This module specifies a very simple API for a lock. A lock is a resource in a multi-
threaded program that can be owned by at most one thread at a time. The API provides
two procedures, Acquire and Release. An Acquire call blocks if another thread owns the
resource. For simplicity, we avoid the need for error returns by specifying that an Acquire
call by the lock’s current owner returns immediately, and a Release call by a thread that
doesn’t own the resource is a no-op.

constant Thread
The set of all thread identifiers.

NoThread ∆= choose nt : nt /∈ Thread
An arbitrary element that is not a thread identifier.

Proc ∆= {“acquire”, “release”}
The set of procedures.

We describe the calling state of the threads by a variable ctl , where ctl [t ] is a record that
describes the state of thread t . The value of ctl [t ] is initially “ready”, and it is changed to
“waiting” when a procedure is called. There is then an internal step, in which t acquires
or releases the lock, that changes ctl [t ] to “done”. The return from the procedure call
resets ctl [t ] to “ready”. While t is executing a procedure call, ctl [t ].proc records which
procedure is being called.

In a more typical API, there would be arguments to procedure calls and values returned.
These would be recorded in additional components of the record ctl [t ]. Most API specifi-
cations are like this one in that the execution of a procedure call is described with a single
internal step (the one that changes ctl [t ] from “waiting” to “done”). However, some API
specs have procedure calls whose description involves performing more than one internal
step. For example, suppose we were specifying a FCFS (first-come, first-served) lock. In
that case, an execution of an Acquire call when the lock is not free might be described
by two internal actions, one that puts the thread onto a waiting queue and another in
which the thread acquires the lock. This might be represented by letting ctl [t ] have two
separate values “waiting1” and “waiting2”, or by having both internal actions occur when
ctl [t ] equals the same value “waiting”.

CtlState ∆=
The set of possible values of ctl [t ] for a thread t .

[state : {“ready”}]
∪ [state : {“waiting”, “done”}, proc : Proc]

variables ctl , owner
The variable owner records the current owner of the lock. It equals NoThread if the
lock is free.

TypeInvariant ∆=
A predicate describing the type-correct values of ctl and owner .

∧ ctl ∈ [Thread → CtlState]
∧ owner ∈ Thread ∪ {NoThread}

Init ∆=
The specification’s initial condition.

∧ ctl = [t ∈ Thread 7→ [state 7→ “ready”]]
∧ owner = NoThread

We now describe the specification’s actions—that is, the top-level disjuncts of the next-
state action.

Call(t , proc) ∆=
Thread t calls procedure proc.

∧ ctl [t ].state = “ready”
∧ ctl ′ = [ctl except ![t ] = [state 7→ “waiting”, proc 7→ proc]]
∧ unchanged owner

Return(t) ∆=
Thread t returns from a procedure call.

∧ ctl [t ].state = “done”
∧ ctl ′ = [ctl except ![t ] = [state 7→ “ready”]]
∧ unchanged owner

DoAcquire(t) ∆=
The internal step in which t acquires ownership of the lock.

∧ ctl [t ].state = “waiting”
∧ ctl [t ].proc = “acquire”
∧ owner ∈ {NoThread , t}
∧ owner ′ = t
∧ ctl ′ = [ctl except ![t ].state = “done”]

DoRelease(t) ∆=
The internal step in which t releases ownership of the lock.

∧ ctl [t ].state = “waiting”
∧ ctl [t ].proc = “release”
∧ owner ′ = if owner = t then NoThread else owner
∧ ctl ′ = [ctl except ![t ].state = “done”]

Internal(t) ∆= DoAcquire(t) ∨DoRelease(t)
For later use, we define Internal(t) to be the disjunction of the internal actions performed
by thread t .

Next ∆=
The complete next-state action.

∃ t ∈ Thread : ∨ ∃ proc ∈ Proc : Call(t , proc)
∨ Return(t)
∨ Internal(t)

We take as the liveness requirement of the API that every thread eventually returns from
any procedure call, unless it is an Acquire call and some other thread acquires the lock and
never releases it. It is a good exercise in understanding fairness and liveness to convince
yourself that this requirement is expressed by conjoining the following fairness condition
to the specification.

Fairness ∆= ∀ t ∈ Thread : SF〈ctl , owner〉(Internal(t) ∨ Return(t))

LockSpec ∆=
The complete specification.

∧ Init
∧2[Next ]〈ctl , owner〉

∧ ∀ t ∈ Thread : SF〈ctl , owner〉(Internal(t) ∨ Return(t))

theorem LockSpec ⇒ 2TypeInvariant

3 The Interface

The lock API specification has two variables, ctl and owner . The variable
owner records the internal state of the lock; the variable ctl describes the call
state of the threads using the API. That is, for each thread t , the value of
ctl [t ] tells if thread t is currently calling the API and, if so, what procedure
it is calling and how far it has progressed in executing the call. If we were
to draw a picture of the specification, it might look like this:

¾ -ctl owner

The picture suggests that the variable ctl describes the interaction of the
semaphore with its environment, and the variable owner describes the in-
ternal state of the semaphore. We regard the value of ctl to be externally
visible, while the value of owner cannot be directly observed—it can at best
be inferred from observations of the sequence of values assumed by ctl .

This picture is actually misleading. If ctl were really just an interface variable,
its value would be changed only by calls and returns to/from procedures. However,
the value of ctl [t ] is changed also by the thread t step that occurs between a call and
a return. Such a step should really be internal, since there is no way for users of the
API to see when that step occurs. (The order in which these steps are performed
for two concurrent procedure calls can at best be inferred from the results returned
by those calls.) By considering ctl to be the interface, we are pretending that there
is a “window” into the API that allows its users to see when the step is performed.
It would be nicer to let ctl model the real interface, so it is changed only by the call
and return actions. This would require introducing an additional internal variable
to remember when an internal step occurs. However, there is no harm in pretending
that the window exists and that users can see when the internal step occurs. Doing
so avoids an extra variable, making the spec a little bit simpler. So we will pretend
that ctl describes the actual interface.

For reasons that will become clear later on, it’s often a good idea to put
into a separate interface module the part of the specification that describes
just the interface. That part of the specification consists of the variable ctl
and everything that involves it alone. We therefore rewrite module Lock1

as two modules, the interface module LockInterface and the module Lock
that extends it and defines the actual API specification, which is formula
LockSpec. Here is module LockInterface

module LockInterface
This module declares the parameters and defines the operators that describe just the
interface of the lock API specification. The first part of this module consists of the
beginning of module Lock1, which contain declarations and definitions that pertain to the
interface.

constant Thread

NoThread ∆= choose nt : nt /∈ Thread

Proc ∆= {“acquire”, “release”}
CtlState ∆=

[state : {“ready”}]
∪ [state : {“waiting”, “done”}, proc : Proc]

variable ctl

For future use, we give names to some conjuncts in the definitions from module Lock1
that mention the interface variable ctl . First, we define IntTypeInvariant and IntInit to
be the conjuncts of TypeInvariant and Init that describe the interface variable ctl .

IntTypeInvariant ∆= ctl ∈ [Thread → CtlState]
IntInit ∆= ctl = [t ∈ Thread 7→ [state 7→ “ready”]]

We now give names to the conjuncts of the actions Call(t , proc) and Return(t) from
module Lock1 that mention the interface variable ctl .

IntCall(t , proc) ∆=
∧ ctl [t ].state = “ready”
∧ ctl ′ = [ctl except ![t ] = [state 7→ “waiting”, proc 7→ proc]]

IntReturn(t) ∆=
∧ ctl [t ].state = “done”
∧ ctl ′ = [ctl except ![t ] = [state 7→ “ready”]]

Module Lock is obtained in a straightforward manner from the part of
module Lock1 not subsumed by the LockInterface module. It begins:

module Lock
extends LockInterface

variable owner

The rest of module Lock is the same as the corresponding part of module
Lock1, except for the definitions of Init , TypeInvariant , Call , and Return. In
those definitions, conjuncts that were given names in module LockInterface
are replaced by those names. For example, the definition of Init becomes

Init ∆=
∧ IntInit
∧ owner = NoThread

and the definition of Return(t) becomes

Return(t) ∆=
∧ IntReturn(t)
∧ unchanged owner

Formula LockSpec, the specification of the lock API, describes the al-
lowed behaviors (sequences of values) of variables ctl and owner . But a user
of the lock API interacts with the API by using the interface. It sees only
the ctl variable, not the owner variable. It cares only about the behavior
of ctl , not of owner . A philosophically correct spec of the API would say
that ctl behaves as if there were a variable owner such that the behaviors
of ctl and owner satisfy formula LockSpec. Such a specification is written
informally as

∃∃∃∃∃∃ owner : LockSpec

For good reasons that do not concern us here, we can’t write the specification
in that way. Instead, the philosophically correct specification PCLockSpec
is defined in the following module PCLock .

module PCLock
extends LockInterface

Inner(owner) ∆= instance Lock
PCLockSpec ∆= ∃∃∃∃∃∃ owner : Inner(owner)!LockSpec

However, we will pretend that PCLockSpec is defined by

PCLockSpec ∆= ∃∃∃∃∃∃ owner : LockSpec

4 Implementation

We now implement the lock API using a trivial version of the bakery algo-
rithm [1]. The bakery algorithm works roughly as follows. A thread that
wants to acquire lock chooses a number, and the thread with the smallest
number gains ownership of the lock. We let num[t ] be the number of thread
t , which initially equals 0. A thread t that wants to acquire the lock sets
num[t ] to 1 plus the largest value of num[u] for every thread u. Our trivi-
alized version of the algorithm has thread t perform the entire operation of
reading the values of all the num[u] and setting num[t ] as a single atomic
step.

We begin by describing the algorithm in vaguely C-like pseudo-code. The
program for thread t is as follows, where initially num[t ] = 0. Statements
are given labels for later use.

Acquire
a: if (num[t] == 0) {

num[t] = 1 + largest num[u] for all threads u}
else {

goto c} ;
b: wait until

(for all threads u : (num[u] = 0) or (num[u] >= num[t]);
c: return ;

Release
a: num[t] = 0 ;
c: return ;

The TLA+ specification of this algorithm is formula BakerySpec of the fol-
lowing module BakeryLock . There is an obvious correspondence between
the specification’s actions and the statements in the pseudo-code above, ex-
cept that the specification has a procedure-call action that does not appear
explicitly in the pseudo-code.

module BakeryLock
extends LockInterface, Naturals

Max (S ) ∆= choose n ∈ S : ∀m ∈ S : n ≥ m
If S is a non-empty set of numbers, then Max (S) is its maximum element.

variables pc, num
pc[t ] equals the “program counter” of thread t—that is, the label of the next statement
to be executed by t . When thread t is not executing a procedure call, pc[t ] equals “a”.

7



TypeInvariant ∆= The type invariant predicate

∧ IntTypeInvariant
∧ pc ∈ [Thread → {“a”, “b”, “c”}]
∧ num ∈ [Thread → Nat ]

Init ∆= The initial predicate

∧ IntInit
∧ pc = [t ∈ Thread 7→ “a”]
∧ num = [t ∈ Thread 7→ 0]

GoTo(t , c1, c2) ∆=
An action expression asserting that control in thread t moves from label c1 to label c2.

∧ pc[t ] = c1
∧ pc′ = [pc except ![t ] = c2]

Below are the algorithm’s actions—that is, the disjuncts of the next-state action. The
value of ctl [t ].state changes from “waiting” to “done” when thread t reaches the return
statement (labeled c) in either procedure. Executing that return statement changes
ctl [t ].state to “ready”.

Call(t , proc) ∆=
The action of thread t calling procedure proc.

∧ IntCall(t , proc)
∧ unchanged 〈pc, num〉

AcqStepA(t) ∆=
The action of thread t executing statement a of the Acquire procedure. The first two
conjuncts assert that t is inside a call of Acquire. Because of the first conjuncts of the
THEN and ELSE parts of the IF formula, the action is enabled only if ctl [t ] = “a”. In
that case, the IF condition is false (num[t ] is greater than 0) iff thread t already owns
the lock.

∧ ctl [t ].state 6= “ready”
∧ ctl [t ].proc = “acquire”
∧ if num[t ] = 0

then ∧GoTo(t , “a”, “b”)
∧ num ′ = [num except

![t ] = if @ = 0 then 1 + Max ({num[u] : u ∈ Thread})
else @]

∧ unchanged ctl
else ∧GoTo(t , “a”, “c”)

∧ ctl ′ = [ctl except ![t ].state = “done”]
∧ unchanged num

AcqStepB(t) ∆=
The action of thread t executing statement b of the Acquire procedure. The first conjunct
implies that pc[t ] equals “b”, which is possible only if t is calling Acquire.

∧GoTo(t , “b”, “c”)
∧ ∀ u ∈ Thread : (num[u] = 0) ∨ (num[u] ≥ num[t ])
∧ ctl ′ = [ctl except ![t ].state = “done”]
∧ unchanged num

Return(t) ∆=
The action of thread t executing the return step (which has label c) in either the Acquire
or Release procedure.

∧GoTo(t , “c”, “a”)
∧ IntReturn(t)
∧ unchanged num

RelStepA(t) ∆=
The action of thread t executing statement a of the Release procedure. The first two
conjuncts assert that t is inside a call of Release.

∧ ctl [t ].state 6= “ready”
∧ ctl [t ].proc = “release”
∧GoTo(t , “a”, “c”)
∧ num ′ = [num except ![t ] = 0]
∧ ctl ′ = [ctl except ![t ].state = “done”]

ImplAction(t) ∆=
For convenience, we define ImplAction(t) to be the disjunction of all of thread t ’s actions
that are performed by the lock implementation itself—which means every action except
the procedure call, which is performed by the user of the API.

∨AcqStepA(t)
∨AcqStepB(t)
∨ Return(t)
∨ RelStepA(t)

Next ∆=
The next-state action.

∃ t ∈ Thread : ∨ ∃ proc ∈ Proc : Call(t , proc)
∨ ImplAction(t)

BakerySpec ∆=
The complete specification. The liveness condition is weak fairness on the implementa-
tion action of every thread.

∧ Init
∧2[Next ]〈ctl , pc,num〉

∧ ∀ t ∈ Thread : WF〈ctl , pc,num〉(ImplAction(t))

theorem BakerySpec ⇒ 2TypeInvariant

I claim that the algorithm specified by formula BakerySpec implements the
lock API. This means that BakerySpec implies the specification PCLockSpec
of the lock API—the specification with owner hidden. In other words, the
formula

BakerySpec ⇒ PCLock

should be valid. This can be asserted in module BakeryLock as follows:

Spec ∆= instance PCLock
theorem BakerySpec ⇒ Spec!PCLock

Instead of instantiating module PCLock , module BakeryLock could extend
it—assuming there were no name conflicts. But instantiating it with renam-
ing in this way works even if there are name conflicts.

Unfortunately, TLC cannot check the correctness of this theorem. Re-
membering the definition of PCLockSpec, this theorem is equivalent to

BakerySpec ⇒ ∃∃∃∃∃∃ owner : LockSpec

and TLC does not handle the hiding operator ∃∃∃∃∃∃ . To figure out how to get
TLC to check correctness of the implementation, we must examine what it
means for BakerySpec to imply (or implement) PCLockSpec.

Let the set Threads of threads be {t1, t2}, and let the operators :> and
@@ be defined as in the standard TLC module so that

f ∆= (t1 :> 3 @@ t2 :> 17)

defines f to be the function with domain {t1, t2} such that f [t1] = 3 and
f [t2] = 17.

Spec BakerySpec implements/implies spec PCLockSpec iff every behavior
that satisfies BakerySpec also satisfies PCLockSpec. Suppose someone gives
us the following behavior that satisfies BakerySpec

Behavior 1



ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 0 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




↓

ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 0 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




↓



ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “b” @@ t2 :> “a”)




↓



ctl = (t1 :> [state 7→ “done”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “c” @@ t2 :> “a”)




↓



ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




↓
. . .

We must show that this behavior satisfies PCLockSpec. How do we do that?
Let’s examine what it means for a behavior to satisfy PCLockSpec. The

only free variable that appears in PCLockSpec is ctl . So, whether or not a
behavior satisfies PCLockSpec depends only on the values that ctl assumes
in that behavior. So, we can forget about the values of mem and pc and ask
if the following behavior, obtained from Behavior 1 by deleting the values
of num and pc, satisfies PCLockSpec.

Behavior 2
[

ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

]

↓
[

ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

]

↓
[

ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

]

↓
[

ctl = (t1 :> [state 7→ “done”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

]

↓
[

ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

]

↓
. . .

By definition, Behavior 2 satisfies PCLockSpec iff we can invent some values
for the variable owner to produce a behavior satisfying LockSpec. We can
do that as follows.

Behavior 3



ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

owner = NoThread




↓



ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

owner = NoThread




↓



ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

owner = NoThread




↓



ctl = (t1 :> [state 7→ “done”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

owner = t1




↓

ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

owner = t1




↓
. . .

It’s easy to check that this behavior satisfies formula LockSpec. (Remember
that all TLA+ specifications allow stuttering steps, so LockSpec allows steps
that leave ctl and owner unchanged.) Therefore Behavior 1, which satisfies
BakerySpec, satisfies PCLockSpec.

To show that BakerySpec implies PCLockSpec, we have to show that
every behavior satisfying BakerySpec also satisfies PCLockSpec. Suppose we
could define an expression owner in terms of the variables ctl , num, and pc
such that the value of owner in each state of a behavior provided the values
of owner needed to show that the behavior satisfies LockSpec. For example,
the value of owner in each of the states of Behavior 1 would be as follows:

Behavior 1



ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 0 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




owner = NoThread

↓



ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 0 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




owner = NoThread

↓



ctl = (t1 :> [state 7→ “waiting”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “b” @@ t2 :> “a”)




owner = NoThread

↓

ctl = (t1 :> [state 7→ “done”, proc 7→ “acquire”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “c” @@ t2 :> “a”)




owner = t1

↓



ctl = (t1 :> [state 7→ “ready”] @@
t2 :> [state 7→ “ready”])

num = (t1 :> 1 @@ t2 :> 0)
pc = (t1 :> “a” @@ t2 :> “a”)




owner = t1

↓
. . .

In other words, if in Behavior 1 we let the value of the variable owner always
equal the value of the expression owner , we get a behavior that satisfies
LockSpec. An equivalent way of saying this is that Behavior 1 satisfies the
formula LockSpec obtained by substituting the expression owner for the
variable owner . If this is true not just of Behavior 1, but of every behavior
satisfying BakerySpec, then BakerySpec implies PCLockSpec.

We can therefore prove that BakerySpec implies PCLockSpec by finding
an expression owner such that every behavior satisfying BakerySpec satis-
fies LockSpec. But every behavior satisfying BakerySpec satisfies LockSpec
iff BakerySpec implies LockSpec. Thus, to prove that BakerySpec implies
PCLockSpec, it suffices to find an expression owner such that BakerySpec ⇒
LockSpec is a valid theorem, where LockSpec is the formula obtained by sub-
stituting owner for owner in LockSpec.

To find a suitable expression owner , we observe that in our simplified
bakery algorithm, a thread t owns the lock iff num[t ] > 0 and t is not waiting
at statement b of the Acquire procedure. In terms of our TLA+ specification,
this means that t owns the lock iff the state predicate IsOwner(t), defined
as follows, is true.

IsOwner(t) ∆= ∧ num[t ] 6= 0
∧ ¬ ∧ ctl [t ].state = “waiting”

∧ ctl [t ].proc = “acquire”
∧ pc[t ] = “b”

We can then define owner to equal

if ∃ t ∈ Thread : IsOwner(t)
then choose t ∈ Thread : IsOwner(t)
else NoThread

This is all expressed in TLA+ in the following module, where owner is
written ownerBar , and LockSpec becomes Bar !LockSpec.

module BakeryLockCorrect
extends BakeryLock

ownerBar ∆= We write ownerBar instead of owner

let IsOwner(t) ∆= ∧ num[t ] 6= 0
∧ ¬ ∧ ctl [t ].state = “waiting”

∧ ctl [t ].proc = “acquire”
∧ pc[t ] = “b”

in if ∃ t ∈ Thread : IsOwner(t)
then choose t ∈ Thread : IsOwner(t)
else NoThread

Bar ∆= instance Lock with owner ← ownerBar
For every operator or formula F defined in module Lock , this statement causes Bar !F to
be defined to equal the operator or formula obtained from F by substituting ownerBar
for owner . Hence, Bar !LockSpec equals LockSpec

theorem BakerySpec ⇒ Bar !LockSpec

TLC can check this theorem.

5 Combining Specifications

Suppose we are writing a specification SysSpec of some system that uses one
or more locks. For concreteness, suppose it uses two locks A and B . We can
picture the system as follows, where ctlA and ctlB are variables representing
the interface between the rest of the system and the two lock APIs.

rest
of

system

¾ -ctlB lock API B

¾ -ctlA lock API A

We expect the two lock APIs to be described by our lock API specification,
so we expect two “copies” of that specification to appear in the definition of
SysSpec, one with ctlA substituted for ctl and the other with ctlB substituted
for ctl .

The philosophically correct specification of a single lock API is formula
PCLockSpec of module PCLock . As explained in Chapter 10 of the TLA+

book, the philosophically correct specification of the system would therefore
have the form

RestOfSys ∧ PCLockSpecA ∧ PCLockSpecB

where PCLockSpecA and PCLockSpecB are formulas PCLockSpec with ctl
instantiated by ctlA and ctlB , respectively, and RestOfSys specifies the rest
of the system. More precisely, formula PCLockSpecA would be defined by

PCLockA ∆= instance PCLock with ctl ← ctlA

PCLockSpecA ∆= PCLockA!PCLockSpec

and PCLockSpecB would be defined analogously.
There are two problems with this philosophically correct approach. The

first is that writing RestOfSys is somewhat tricky. The second is that TLC
can’t handle this kind of “compositional” specification. So, we’ll do it an
easier way, using two copies of module Lock instead of module PCLock .

As an example, let’s suppose we are modeling a multi-threaded program
that uses two locks, which we call locks A and B . Suppose the program for
each thread contains the following piece of code which might arise if x and
y are shared variables protected by locks A and B , respectively:

...
a : Acquire Lock A
b : Acquire Lock B
c : x = x + y
d : Release Lock B
e : Release Lock A

...

To model this program in TLA+, we use variables x and y to represent the
program variables x and y, and represent program control with a variable
pc—as we did above in module BakeryLock . Letting Thread denote the set
of threads, we might begin the specification as follows. (The . . . represents
the other specification variables, including ones representing other program
variables.)

extends Naturals
constant Thread
variables pc, x , y , . . .

We would “import” two copies of the Lock module, with different instantia-
tions of that module’s variables. We represent lock A with a copy in which
ctlA is substituted for ctl and ownerA is substituted for owner ; and similarly
for lock B .

variables ctlA, ownerA, ctlB , ownerB
LockA ∆= instance Lock with ctl ← ctlA, owner ← ownerA
LockB ∆= instance Lock with ctl ← ctlB , owner ← ownerB

The other parameter of module Lock , the constant Thread , is instantiated
by the constant Thread of the current module. For convenience, we make
the following definitions:

aVars ∆= 〈ctlA, ownerA〉
bVars ∆= 〈ctlB , ownerB〉

Our specification uses definitions from the two instances of the Lock module
in the initial predicate, the type invariant, and the next-state action. The
initial predicate is:

Init ∆= ∧ LockA!Init
∧ LockB !Init
∧ pc = [t ∈ Thread 7→ . . . ]
∧ x = . . .
∧ y = . . .
. . .

The first two conjuncts specify the initial values of the variables ctlA, ownerA,
ctlB , and ownerB . Similarly, the type invariant is:

TypeInvariant ∆= ∧ LockA!TypeInvariant
∧ LockB !TypeInvariant
∧ pc ∈ . . .
. . .

We now examine how we represent the program statements that acquire and
release the locks—for example, the statement:

a : Acquire Lock A

17



In our model, an execution of this statement by a thread t consists of three
steps: the call of the acquire procedure, the internal step of the lock in
which control changes from “waiting” to “done”, and the return. The call
and return are described by the following actions, where again “. . .” stands
for the additional variables of the specification:

StepACall(t) ∆=
∧ pc[t ] = “a”
∧ LockA!Call(t , “acquire”)
∧ unchanged 〈pc, x , y , . . . , bVars〉

StepAReturn(t) ∆=
∧ pc[t ] = “a”
∧ LockA!Return(t)
∧ pc′ = [pc except ![t ] = “b”]
∧ unchanged 〈x , y , . . . , bVars〉

Note that we must specify that the variables x , y , . . . , ctlB , and ownerB
are left unchanged. (Leaving varsB unchanged is equivalent to leaving both
ctlB and ownerB unchanged.) The internal steps for all of thread t ’s lock
A procedure calls will be described later by a single action.

The acquiring of lock B by program statement b is handled similarly.
Execution of the statement

c : x = x + y

would probably be modeled as a single step satisfying the action

StepC (t) ∆=
∧ pc[t ] = “c”
∧ x ′ = x + y
∧ pc′ = [pc except ![t ] = “d”]
∧ unchanged 〈y , . . . , aVars, bVars〉

The statements d and e would be represented similarly. For example, state-
ment d would be described by the two actions

StepDCall(t) ∆=
∧ pc[t ] = “d”
∧ LockB !Call(t , “release”)
∧ unchanged 〈pc, x , y , . . . , aVars〉

StepDReturn(t) ∆=

∧ pc[t ] = “d”
∧ LockB !Return(t)
∧ pc′ = [pc except ![t ] = “e”]
∧ unchanged 〈x , y , . . . , aVars〉

The next-state action would be as follows, where the last two disjuncts
describe the internal steps performed by all of thread t ’s calls to the lock
procedures.

Next ∆=
∃ t ∈ Thread : ∨ StepACall(t)

∨ StepAReturn(t)
. . .
∨ ∧ LockA!Internal(t)
∧ unchanged 〈pc, x , y , . . . , bVars〉

∨ ∧ LockB !Internal(t)
∧ unchanged 〈pc, x , y , . . . , aVars〉

Those two disjuncts would probably also appear in any fairness require-
ments.

TLC can handle this specification. However, getting it to do so requires
one non-obvious trick. Recall that module LockInterface has the definition

NoThread ∆= choose nt : nt /∈ Thread

which TLC cannot handle. Hence, one must tell TLC to assign a model value
to the constant NoThread . (We usually assign the model value NoThread
to this constant.) Such an assignment is usually performed by putting the
statement

NoThread = NoThread

into the CONSTANTS part of the configuration file. However, that doesn’t
work here because NoThread is defined in an instantiated (rather than an
extended) module. Instead, you can use the following statement in the
configuration file:

NoThread =[LockInterface] NoThread

which tells TLC to make the assignment in the LockInterface module. Equiv-
alently, you can instead tell TLC to make the assignment in the Lock module
with the assignment

NoThread =[Lock] NoThread

6 A Simpler Lock Specification

Consider again the example multi-threaded program of Section 5, which
contained the following piece of code.

...
a : Acquire Lock A
b : Acquire Lock B
c : x = x + y
d : Release Lock B
e : Release Lock A

...

We represented each of the Acquire and Release statements by two actions.
In addition, the next-state action contained two disjuncts describing the
internal steps of each of the locks. In many cases, we would like to model an
execution of Acquire or Release as a single step. In addition to simplifying
the specification, reducing the execution of each procedure call from three
steps to one step would decrease the size of the state space, making model
checking easier.

We could obtain this kind of simple model of the program by starting
with a simpler model of a lock. The following module specifies such a simpler
model; it is simple enough that it should require no explanation.

module SimpleLock
constant Thread
NoThread ∆= choose nt : nt /∈ Thread

variable owner

Init ∆= owner = NoThread
TypeInvariant ∆= owner ∈ Thread ∪ {NoThread}
Acquire(t) ∆= ∧ owner ∈ {NoThread , t}

∧ owner ′ = t

Release(t) ∆= ∧ owner ′ = if owner = t then NoThread else owner

Next ∆= ∃ t ∈ Thread : Acquire(t) ∨ Release(t)

SimpleLockSpec ∆= Init ∧2[Next ]owner

The first part of our specification of the program that uses the two locks A
and B would be just like the one in Section 5:

extends Naturals
constant Thread
variables pc, x , y , . . .

variables ctlA, ownerA, ctlB , ownerB
LockA ∆= instance SimpleLock with ctl ← ctlA, owner ← ownerA
LockB ∆= instance SimpleLock with ctl ← ctlB , owner ← ownerB

Init ∆= ∧ LockA!Init
∧ LockB !Init
. . .

TypeInvariant ∆= ∧ LockA!TypeInvariant
∧ LockB !TypeInvariant
. . .

However, the descriptions of the program statements would be simpler. For
example, statement a would be described by the single action

StepA(t) ∆=
∧ pc[t ] = “a”
∧ LockA!Acquire(t)
∧ pc′ = [pc except ![t ] = “b”]
∧ unchanged 〈x , y , . . . , ownerB〉

There would be no extra disjuncts of the next-state relation, since the simple
lock specification has no extra internal steps.

Which is the best specification of a lock, the one in module Lock or the
simple one of module SimpleLock? Neither. There is no “best” specification
of any system. A specification is written for a purpose. What kind of spec-
ification you write depends on that purpose. Module Lock was written to
specify an API. Module SimpleLock doesn’t provide much of a specification
of a lock API. In fact, formula Spec of module SimpleLock is equivalent to
the formula

∧ owner = NoThread
∧2 [ ∨ ∧ owner = NoThread

∧ owner ′ ∈ Thread
∨ ∧ owner ∈ Thread
∧ owner ′ = NoThread

]owner

Both formulas describe the same possible sequences of changes to the vari-
able owner .1 This equivalent way of writing it shows that the specification
of module SimpleLock isn’t a very good one for explaining how to use a lock.

If one writes two different specifications of the same system for differ-
ent purposes, there will most likely not be any simple formal relation be-
tween those two specifications. In this particular example, specification
LockSpec of module Lock implements specification SimpleLockSpec of mod-
ule SimpleLock . (Such an implementation relation will hold whenever we
simplify an API specification by eliminating the calls and returns and just
leaving the internal steps.) However, the two specifications of the program
that uses the locks would be different. For example, in the one we described
in Section 5, the value of ownerA and pc[t ] change in separate steps during
the execution of statement a; in the specification sketched in this section,
the StepA(t) action changes ownerA and pc[t ] in a single step.

Although it does not yet happen often in industrial applications, you may
sometimes want a specification to serve two functions. In our toy example,
we wanted to use the lock API’s specification both to check an implemen-
tation of the API and to check a program that uses the API. It’s nice to
check the program using the same specification of the API that we verify
is satisfied by the API’s implementation. Using two different specifications
would introduce a possible source of errors—namely, that the program relies
on some property of the API that isn’t satisfied by the implementation. On
the other hand, using the more accurate specification of module Lock leads
to a larger state space, making model checking less effective. This could
cause us to miss an error that we would have found had we checked larger
instances of the system with the simpler lock specification.

Deciding whether to use the same specification for different purposes or
to write different specifications requires engineering judgment.

